PharmaSUG 2017 - Paper DV10

Automated DSMB Presentation in SAS®: Yeah, I’d Submit That! How to auto populate PowerPoint presentations so you don’t have to

Brett Jepson and Kaitie Lawson, Rho Inc., Chapel Hill, NC

ABSTRACT

Data & Safety Monitoring Board (DSMB) members are increasingly requesting high-level overview presentations to assess the current status of a study in lieu of lengthy binders. When multiple studies report to the same DSMB, a certain level of standardization across presentations is desirable to facilitate ease in understanding and familiarity with the presentation format for the DSMB members. Manually entering text and data into a presentation is tedious and is subject to transcription errors. By utilizing ODS POWERPOINT, the ODSTEXT and ODSLIST procedures, and a series of other familiar SAS® procedures, we have created a suite of SAS macros that automatically populate presentations, therefore eliminating the need to manually enter data. These SAS macros allow for presentations conforming to a general template to be created in a standardized manner which saves time, ensures data accuracy and integrity, and provides continuity in data presentation to DSMB members.

INTRODUCTION

Data & Safety Monitoring Board (DSMB) members are increasingly requesting high-level overview presentations to assess the current status of a study in lieu of reviewing lengthy binders. The need for succinct and informative slides, tables, and graphics are imperative so DSMB members can obtain the information that they need to make decisions about how (or if) a study should continue. Traditionally, data and information from summary tables and displays would need to be entered into a presentation manually. However, manually entering text and data into a presentation is tedious and is subject to transcription errors. We have created one SAS program based on multiple macro calls that creates a Microsoft PowerPoint presentation that adheres to our established template and automatically populates all the required slides by utilizing ODS POWERPOINT, the ODSTEXT and ODSLIST procedures, as well as a series of other familiar SAS procedures. This paper, which assumes a basic level of SAS macro coding, will show examples of how one program can be created to produce all the slides necessary for a DSMB presentation by automatically populating an existing template.

ODS POWERPOINT OVERVIEW

The ODS PowerPoint destination has been available in SAS 9.4 since 2013 which allows users to format SAS procedure output as PowerPoint slides. Accessing the destination for PowerPoint is similar to accessing other ODS destinations by simply specifying the POWERPOINT keyword in the ODS statement in conjunction with the FILE= option:

ODS POWERPOINT FILE="Example.pptx";

<Additional SAS Code>

ODS POWERPOINT CLOSE;

There are essentially three types of data we want to incorporate into PowerPoint slides for DSMB reports: text, tables and figures. We will highlight how to produce these slides using the REPORT, SGPLOT, ODSTEXT and ODSLIST procedures in conjunction with the ODS POWERPOINT destination. After covering these basic procedures, we will show how an entire DSMB PowerPoint presentation can be created from one single program and how to create SAS macros so that SAS programs generating DSMB presentations can accommodate multiple studies.
TEXT SLIDES USING THE ODS TEXT AND ODS LIST PROCEDURES

All PowerPoint slides contain some form of text. When we say 'Text Slide', we are mainly referring to title slides or slides that contain bulleted/summary information without any tables or figures. For example, Figure 1 and Figure 2 display typical text slides that you may see at the beginning of any DSMB report. Figure 1 is a title slide, and Figure 2 is a bulleted slide describing the study design.

Title of the DSMB Presentation

Protocol Number: XXXX-01
Sponsor: Pharma Company
Report Prepared by:
My CRO, Inc. (My City, USA)
Data Cutoff Date: March 30, 2017
DSMB Meeting Date: May 15, 2017
Open Meeting Presentation

Figure 1. Example of a Title Slide

Study Design

• Phase 3, 5-site, 2-arm, double-blind, placebo-controlled, randomized trial in participants with allergic rhinitis
  • Arm 1: Sublingual-Immunotherapy
  • Arm 2: Placebo
• Enrollment goal: 100 participants randomized
• Primary Endpoint: Change in FEV1 % predicted from baseline

Figure 2. Example of a Bulleted List Slide

It might not be intuitive that creating text slides in SAS would save time; admittedly, creating text slides is fairly simple in PowerPoint. However, when we create text slides in ODS POWERPOINT, we allow for the possibility to create the entire presentation within one program. Otherwise, text slides would either need to be hand entered after using ODS POWERPOINT to create other slides with tables or graphics, or copied and pasted from a previous presentation. Either of the two methods introduces potential typographical errors. In addition, since DSMB presentations often repeated at regular intervals, including code to create all slides of a subsequent presentation within a single SAS program allows for easy reruns of the program with little modification.
EXAMPLE 1: TITLE SLIDE USING ODSTEXT

The code to produce the slide in Figure 1, which uses the ODSTEXT procedure, is below:

```sas
ODS ESCAPECHAR='~';
ODS POWERPOINT FILE="TITLE_SLIDE.pptx" NOGTITLE NOGFOOTNOTE;
PROC ODSTEXT;
   P " " / STYLE={FONTSIZE=30};
   P "Title of the DSMB Presentation" / STYLE={FONTSIZE=20 JUST=C FONTWEIGHT=BOLD};
   P "Protocol Number: " / STYLE={FONTSIZE=12 JUST=C};
   P "Sponsor: Pharma Company" / STYLE={FONTSIZE=12 JUST=C};
   P "Report Prepared by: My CRO, Inc. (My City, USA)" / STYLE ={FONTSIZE=8 JUST=C};
   P "Data Cutoff Date: March 30, 2017 DSMB Meeting Date: May 15, 2017 Open Meeting Presentation" / STYLE ={FONTSIZE=8 JUST=C};
RUN;
ODS POWERPOINT CLOSE;
```

Each P statement creates a new text block, and there are a variety of options that can be included as STYLE elements to change the orientation and font of text. Two blank paragraphs are included to add spacing above and below the title.

The "~n" code creates a hard return and moves all subsequent text to the next line. The escape character, which is defined in the ODS ESCAPECHAR statement, was chosen to be '~' in our code, but could be set to any acceptable escape character allowable in SAS. This method allows for text to be grouped in the same text box in the PowerPoint slide, while including text in a new P statement creates a new text box. The amount of space between text boxes will generally be larger than the space between lines within a text box.

Below is an example of how to Macroitize the preceding code:

```sas
%MACRO TITLESLIDE (PRESENTATION_TITLE=,PROTOCOL_NUMBER=,SPONSOR=,CUTOFF_DATE=,DSMB_DATE=);
ODS POWERPOINT FILE="TITLE_SLIDE.pptx" NOGTITLE NOGFOOTNOTE;
PROC ODSTEXT;
   P " " / STYLE={FONTSIZE=30};
   P "&PRESENTATION_TITLE." / STYLE={FONTSIZE=20 JUST=C FONTWEIGHT=BOLD};
   P " " / STYLE={FONTSIZE=30};
   P "Protocol Number: &PROTOCOL_NUMBER.~n Sponsor: &SPONSOR." / STYLE={FONTSIZE=12 JUST=C};
   P "Report Prepared by:~n My CRO, Inc. (My City, USA)" / STYLE ={FONTSIZE=8 JUST=C};
   P "Data Cutoff Date: &CUTOFF_DATE.~n DSMB Meeting Date: &DSMB_DATE.~n Open Meeting Presentation" / STYLE ={FONTSIZE=8 JUST=C};
RUN;
%MEND TITLESLIDE;

%TITLESLIDE (PRESENTATION_TITLE= Title of the DSMB Presentation,PROTOCOL_NUMBER= XXXX-01,SPONSOR= Pharma Company,CUTOFF_DATE= %STR(March 30, 2017),DSMB_DATE= %STR(May 15, 2017))
```

The %STR macro function is used to treat the value as text only. For example, when a value is a date with a comma, as in the last two macro variables (CUTOFF_DATE and DSMB_DATE), %STR masks the
Automated DSMB Presentation in SAS®: Yeah, I’d Submit That! How to auto populate PowerPoint presentations so you don’t have to, continued

commas. Without using the %STR function, the macro will assume that the clause after the comma (in this case, the year) is a new macro variable definition and will give a WARNING statement concerning positional parameters. Other macro functions that are similar to %STR can also be used, depending on which special characters are needed to be masked (e.g. %BQUOTE and %NBRQUOTE).

EXAMPLE 2: STUDY DESIGN SLIDE USING ODSL

The code to produce the slide in Figure 2, which uses the ODSLIST procedure, is below:

```sas
ODS POWERPOINT FILE = "DESIGN_SLIDE.PPTX";
PROC ODSLIST;
  ITEM;
    P "Phase 3, 5-site, 2-arm, double-blind, placebo-controlled, randomized trial in participants with allergic rhinitis" || " " /STYLE={FONTSIZE=12 LISTSTYLETYPE= "disc"};
    LIST / STYLE={FONTSIZE=10.5 LISTSTYLETYPE= "circle"};
    ITEM "Arm 1: Sublingual-Immunotherapy";
    ITEM "Arm 2: Placebo";
  END;
END;
ITEM "Enrollment goal: 100 participants randomized" || " " /STYLE={FONTSIZE=12 LISTSTYLETYPE= "disc"};
ITEM "Primary Endpoint: Change in FEV1 % predicted from baseline" /STYLE={FONTSIZE=12 LISTSTYLETYPE= "disc"};
RUN;
ODS POWERPOINT CLOSE;
```

The ODSLIST procedure uses ITEM blocks to create lists, which can be customized using LISTSTYLETYPE in its appearance (numbered, bulleted, etc.). In addition, sub-lists can be created within an ITEM. This is done by including a set of ITEM statements under a LIST statement.

If we create macro variables to generate the key elements of the design slide (i.e. Design, Arms, Enrollment Goal, and Primary Endpoint), we could have one program that would create this exact same slide for any number of studies which in turn provides efficiency and standardization across all presentations made to DSMB members. The following example provides the macro code and two different macro calls. Figure 3 shows the two different design slides that are produced from the two individual macro calls.

**Study Design**

- Phase 2, 3-site, 2-arm, double-blind, placebo-controlled, randomized trial in adults aged 18-55 with Lupus Nephritis
  - Arm 1: A
  - Arm 2: B
- Enrollment goal: 100 Randomized
- Primary Endpoint: Proportion of subjects who achieve a complete response at 24 weeks

**Study Design**

- Phase 3, 1-site, 3-arm, double-blind, double-dummy, placebo-controlled, randomized trial in adults aged 18-55 with Seasonal Allergic Rhinitis
  - Arm 1: Subcutaneous Immunotherapy + Sublingual Placebo
  - Arm 2: Sublingual Immunotherapy + Subcutaneous Placebo
  - Arm 3: Double Placebo
- Enrollment goal: 300 randomized in a 1:1:1 fashion
- Primary Endpoints:
  - Total Nasal Symptom Score Area Under the Curve in response to Nasal Allergen Challenge measured at 3 Years
  - Peak Nasal Inspiratory Flow Area Under the Curve in response to Nasal Allergen Challenge measured at 3 Years

Figure 3. Study Design Slides Produced from Two Separate Macro Calls
The following macro and subsequent macro calls create both of the slides presented in Figure 3:

```sas
%MACRO STUDYDESIGN
(DESIGN=,ARM1=,ARM2=,ARM3=,ARM4=,ARM5=,ARM6=,ENROLLGOAL=,PRIMARY=,
 PRIMARY2=,PRIMARY3=,PRIMARY4=,FOOTNOTE=);
TITLE "Study Design";
FOOTNOTE;
%IF &FOOTNOTE. ^= %THEN FOOTNOTE J=L "&FOOTNOTE";;
ODS ESCAPECHAR='~';
PROC ODSLIST;
ITEM;
P "&DESIGN." %IF &ARM1. = %THEN || "~n "; / STYLE={FONTSIZE=12
LISTSTYLETYPE="disc"};
%IF &ARM1. ^= %THEN %DO;
LIST / STYLE={FONTSIZE=10.5 LISTSTYLETYPE="circle"};
ITEM "Arm 1: &ARM1.";
%IF &ARM2. ^= %THEN ITEM "Arm 2: &ARM2.";
%IF &ARM3. ^= %THEN ITEM "Arm 3: &ARM3.";
%IF &ARM4. ^= %THEN ITEM "Arm 4: &ARM4.";
%IF &ARM5. ^= %THEN ITEM "Arm 5: &ARM5.";
%IF &ARM6. ^= %THEN ITEM "Arm 6: &ARM6." || "~n "; ;;
END;
%END;
END;
ITEM "Enrollment goal: &ENROLLGOAL.~n || " /STYLE={FONTSIZE=12
LISTSTYLETYPE="disc"};
%IF &PRIMARY2. = %THEN %DO;
ITEM "Primary endpoint: &PRIMARY. /STYLE={FONTSIZE=12
LISTSTYLETYPE="disc"};
%END;
%ELSE %DO;
ITEM;
P "Primary endpoints:" / STYLE={FONTSIZE=12 LISTSTYLETYPE="disc"};
LIST / STYLE={FONTSIZE=10.5 LISTSTYLETYPE="circle"};
ITEM "&PRIMARY.";
ITEM "&PRIMARY2.";
%IF &PRIMARY3. ^= %THEN ITEM "&PRIMARY3.";;
%IF &PRIMARY4. ^= %THEN ITEM "&PRIMARY4.";;
END;
%END;
RUN;
%MEND STUDYDESIGN;
```
%STUDYDESIGN(DESIGN= %STR(Phase 2, 3 site, 2 arm, double-blind, placebo-controlled, randomized trial in adults aged 18-55 years with Lupus Nephritis)
,ARM1= A
,ARM2= B
,ENROLLGOAL= 100 Randomized
,PRIMARY= Proportion of subjects who achieve a complete response at 24 weeks);

%STUDYDESIGN(DESIGN= %STR(Phase 3, 1 site, 3 arm, double-blind, double-dummy, placebo-controlled, randomized trial in adults aged 18-55 with Seasonal Allergic Rhinitis)
,ARM1= %STR(Subcutaneous Immunotherapy + Sublingual Placebo)
,ARM2= %STR(Sublingual Immunotherapy + Subcutaneous Placebo)
,ARM3= Double Placebo
,ENROLLGOAL= %STR(300 randomized in a 1:1:1 fashion)
,PRIMARY= Total Nasal Symptom Score Area Under the Curve in response to Nasal Allergen Challenge measured at 3 Years
,PRIMARY2= Peak Nasal Inspiratory Flow Area Under in response to Nasal Allergen Challenge the Curve measured at 3 Years);

The macro code that we have designed is flexible enough to create a Design Slide for a study with up to 6 different arms and up two 4 different primary endpoints. Having these options built into the code allows for a fair amount of customization and flexibility. As you can see, the code used to create these slides is contained within the macro called %STUDYDESIGN. We have also developed a macro called %STUDYDESIGN_OBSERVATIONAL with its own set of parameters in case the study that needs to be reported to the DSMB is observational as opposed to an interventional trial.

TEXT SLIDE HELPFUL HINT #1

ODS LAYOUT may also be helpful in customizing slides to fit your needs since it allows creating multiple lists, tables or figures on the same slide. We found ODS LAYOUT useful in creating a slide containing an outline of the presentation, with two text boxes side by side; the left side containing the slide name, with the right side containing the corresponding slide number. ODS LAYOUT code is structured like this:

ODS LAYOUT GRIDDED COLUMNS=2;

ODS REGION WIDTH=5IN;
<CODE TO CREATE LEFT COLUMN>

ODS REGION WIDTH=3IN;
<CODE TO CREATE RIGHT COLUMN>

ODS LAYOUT END;

TEXT SLIDE HELPFUL HINT #2

For lists that could contain a varying number of items depending on the particular report, we found it useful to create a macro that specified a large number of macro variables as list items. Only specified list items will appear in the slide. An example of a macro that demonstrates this idea is shown here:

%MACRO APPENDIX(LIST1=,LIST2=,LIST3=,LIST4=,LIST5=,LIST6=,LIST7);
ODS POWERPOINT FILE = "APPENDIX.PPTX";
ODS ESCAPECHAR='~';
TITLE "APPENDICIES";
PROC ODSLIST;
    ITEM;
    LIST / STYLE={FONTSIZE=16 LISTSTYLETYPE="UPPER_ALPHA"};
TABLE SLIDES USING THE REPORT PROCEDURE

Table slides refer to PowerPoint slides that include summary data presented in a tabular format. These are the slides where the largest potential for transcription errors could occur and therefore provide the best opportunity to increase efficiencies. Figure 4 displays a table slide summarizing protocol deviations.

Protocol Deviations

<table>
<thead>
<tr>
<th>Deviation</th>
<th>Since Last Report: N = 20</th>
<th>To Date: N = 50</th>
</tr>
</thead>
<tbody>
<tr>
<td>Informed Consent</td>
<td>4 (20%)</td>
<td>8 (16%)</td>
</tr>
<tr>
<td>Missed Safety, Efficacy or Endpoint Assessments</td>
<td>7 (35%)</td>
<td>12 (24%)</td>
</tr>
<tr>
<td>Reporting of SAEs</td>
<td>1 (5%)</td>
<td>4 (8%)</td>
</tr>
<tr>
<td>Study treatment administration</td>
<td>5 (25%)</td>
<td>14 (28%)</td>
</tr>
<tr>
<td>Use of Prohibited Medications</td>
<td>1 (5%)</td>
<td>5 (10%)</td>
</tr>
<tr>
<td>Other: Violation of visit window</td>
<td>2 (10%)</td>
<td>7 (14%)</td>
</tr>
</tbody>
</table>

Figure 4. Protocol Deviations Table

EXAMPLE 3: PROTOCOL DEVIATIONS SUMMARY SLIDE USING PROC REPORT

This table summarizes protocol deviations that subjects have experienced during the entire study (as seen in the 'To Date' column) as well as protocol deviations that have occurred since the last DSMB report (the 'Since Last Report' column). At study start up, each study team defines a set of reasons for documenting protocol deviations. However, more often than not additional deviations occur that are not in the pre-defined list. The macro code that we have developed is flexible enough to summarize these 'Other, specify' deviations as they are also important to report to the DSMB.

The code used to make the Protocol Deviation slide is listed below:

```
ODS POWERPOINT FILE = "DEVIATION SLIDE.PPTX";
TITLE "Protocol Deviations";
PROC REPORT DATA=SUMMARY_DATA NOWD SPLIT = "-" HEADSKIP CENTER
   STYLE(REPORT) = [HEIGHT=6IN WIDTH=9.25IN]
```
Automated DSMB Presentation in SAS®: Yeah, I’d Submit That! How to auto populate PowerPoint presentations so you don’t have to, continued

```
STYLE(HEADER)=[BACKGROUND=BIGB FONTWEIGHT=BOLD FOREGROUND=WHITE];
COLUMN LINE COL1 ("Deviations: n (%)" COL2 COL3);
DEFINE LINE / ORDER NOPRINT;
DEFINE COL1 / DISPLAY LEFT FLOW "~Deviation" STYLE=[JUST=LEFT CELLWIDTH=7IN]
STYLE(COLUMN)=[ASIS=ON] STYLE(HEADER)=[JUST=LEFT FONTWEIGHT=BOLD
FONTSIZE=5.8 FOREGROUND=WHITE BACKGROUND=BIGB];
DEFINE COL2 / DISPLAY LEFT FLOW "Since Last Report:~N = 25"
STYLE=[JUST=CENTER CELLWIDTH=2IN] COLOR=WHITE STYLE(HEADER)=[JUST=CENTER
FONTWEIGHT=BOLD FONTSIZE=5.8 FOREGROUND=WHITE BACKGROUND=BIGB];
DEFINE COL3 / DISPLAY LEFT FLOW "To Date:~N = 38" STYLE=[JUST=CENTER
CELLWIDTH=2IN] COLOR=WHITE STYLE(HEADER)=[JUST=CENTER FONTWEIGHT=BOLD
FONTSIZE=5.8 FOREGROUND=WHITE BACKGROUND=BIGB];
RUN;
ODS POWERPOINT CLOSE;
```

Furthermore, the DSMB might also be interested in seeing how the protocol deviations are distributed across each site in a multi-center study. An example of this type of table is shown in Figure 5.

**Figure 5. Protocol Deviations Table by Site**

<table>
<thead>
<tr>
<th>Site</th>
<th>Deviations: n (%)</th>
<th>Since Last Report: N = 20</th>
<th>To Date: N = 50</th>
</tr>
</thead>
<tbody>
<tr>
<td>Atlanta</td>
<td></td>
<td>0 (0%)</td>
<td>2 (4%)</td>
</tr>
<tr>
<td>Boston</td>
<td></td>
<td>1 (5%)</td>
<td>5 (10%)</td>
</tr>
<tr>
<td>Dallas</td>
<td></td>
<td>12 (60%)</td>
<td>21 (42%)</td>
</tr>
<tr>
<td>Denver</td>
<td></td>
<td>4 (20%)</td>
<td>11 (22%)</td>
</tr>
<tr>
<td>New York</td>
<td></td>
<td>1 (5%)</td>
<td>4 (8%)</td>
</tr>
<tr>
<td>Washington, DC</td>
<td></td>
<td>2 (10%)</td>
<td>7 (14%)</td>
</tr>
</tbody>
</table>

The code to create this table would be similar to the code to create the table in Figure 4. Because of these similarities, creating these two tables within one macro eliminates the amount of code needed to create them, resulting in more efficiency and fewer opportunities to create mistakes. However, single-site studies would not need a by-site table. A macro with the capabilities to allow for a user to decide if this table is included in a presentation meets the needs of all studies.
The following code structure shows how using simple macro code within a macro can make the Protocol Deviations by Site table optional:

```sas
ODS POWERPOINT FILE = "DEVIAITION SLIDE.PPTX";
TITLE "Protocol Deviations";
<CODE TO CREATE A PROTOCOL DEVIATIONS TABLE>
%IF &BYSITE. = Y %THEN %DO;
   TITLE "Protocol Deviations by Site";
   <CODE TO CREATE A PROTOCOL DEVIATIONS BY SITE TABLE>
%END
ODS POWERPOINT CLOSE;
```

The BYSITE macro variable is a variable that would need to be included in the macro variables. When set to Y in the macro call, the macro creates the by-site table. Otherwise, the by-site table is not created.

**TABLE SLIDE HELPFUL HINT #1**

It is helpful if datasets are standardized across the studies that will be using the same DSMB presentation template. Standardized datasets allow for fewer macro variables to be defined and allow default values to be set for macro variables when developing the macro code. Datasets that usually require little customization (e.g. adverse events and protocol deviations) often occupy much of the presentation to DSMBs and are excellent candidates for standardization between studies.

**TABLE SLIDE HELPFUL HINT #2**

When reporting categorical variables in tables where there is an ‘Other, specify’ option, an easy way to distinguish the ‘Other, specify’ items in the table is to create a shell of all options in the pre-defined list, while adding a prefix of ‘Other:’ to deviations not in the pre-defined list. Here is some code that performs this task, which would be inserted in the middle of a macro:

```sas
/*Create a macro variable that counts the number of deviation types*/
DATA _NULL_
   TYPES = "&TYPESLIST.";
   NUMTYPES = COUNT(TYPES,')') + 1;
   CALL SYMPUTX("NUMTYPES",NUMTYPES);
RUN;

/*Create a shell with 1 record per deviation type*/
DATA SHELL;
   FORMAT COL1_ $132.;
   TYPES = "&TYPESLIST.";
   %DO I = 1 %TO &NUMTYPES.;
      COL1_ = SCAN(TYPES,&I.,')');
      SORT = &I.;
      OUTPUT;
   %END;
RUN;

PROC SORT DATA = SHELL;
   BY COL1_;
RUN;

/*Merge deviation dataset with the shell and append "Other:" to anything that is not in the list of case report form-specified deviations. Also, sort the "Other" types last so they appear at the end of the list in the table*/
DATA DEVIATIONS;
```

9
Automated DSMB Presentation in SAS®: Yeah, I’d Submit That! How to auto populate PowerPoint presentations so you don’t have to, continued

```
LENGTH COL1 $132;
MERGE DEVIATION_DATASET SHELL (IN=A);
BY COL1 ;
IF A THEN COL1 = COL1 _;
ELSE DO;
   COL1 = "Other: " || COL1 _;
   SORT = 99;
END;
RUN;
```

This method ensures that no deviations are left out inadvertently, as they are classified with the ’Other’ reason for deviations. The corresponding macro call would use a pipe to separate the reasons for deviation and looks something like this:

```
%DEVIATION_TABLE(...,TYPELIST=%STR(REASON FOR DEVIATION 1|REASON FOR DEVIATION 2|...))
```

**TABLE SLIDE HELPFUL HINT #3**

For an initial DSMB meeting, the column for “Since Last Report” won’t be needed as there have been no previous reports, so it is prudent to include an option that will include or exclude this column. Similar coding as the code to include a by-site table (see Example 3) can be used to include or exclude a column in the REPORT procedure. It is necessary to derive a variable that determines which data have occurred since the last report, but that can be done by simply including a date that will act as a cut off in your macro coding.

**FIGURE SLIDES USING THE SGPLOT PROCEDURE**

Figure slides have the ability to convey complex information about the status of a trial in a succinct manner. One of the key aspects of a trial that the DSMB likes to track is enrollment over time. If study enrollment is too slow, this might mean that the enrollment criteria are too strict. A slow enrollment will also have an effect on the budget for the study. If study enrollment seems to be taking longer than expected, the DSMB might suggest that the inclusion and exclusion criteria be adjusted. An example of a Figure Slide that displays enrollment over time is displayed in Figure 6.
Automated DSMB Presentation in SAS®: Yeah, I'd Submit That! How to auto populate PowerPoint presentations so you don’t have to,
continued

Enrollment By Study Month

![Enrollment By Study Month](image)

Figure 6. Enrollment over Time Figure

The code used to create this slide is below:

```sas
ODS POWERPOINT FILE = 'ENROLLMENT SLIDE.PPTX';
PROC SGPLOT DATA = PLOT NOAUTOLEGEND;
   SERIES X=PRO Y=SUBSP / LINEATTRS=(COLOR=BLACK) LEGENDLABEL="Expected"
      NAME="e";
   STEP X=MONTH Y=SUBS / LINEATTRS=(COLOR=BLUE) CURVELABEL = "N= 205 at Month 12"
      CURVELABELATTRS=(SIZE=9PT) LEGENDLABEL="Actual" NAME="a";
   XAXIS LABEL = "Study Month" OFFSETMAX=0.25;
   YAXIS LABEL = "Number Randomized" OFFSETMAX=.05;
   KEYLEGEND "e" "a" / NOBORDER LOCATION=INSIDE POSITION=TOPLEFT;
   TITLE "Enrollment By Study Month" BOLD H=35PT;
RUN;
ODS POWERPOINT CLOSE;
```

Just like with the text slides, it might not be intuitive that using ODS POWERPOINT and PROC SGPLOT would save time when creating figure slides for a presentation. However, if we create macro variables to populate a few key elements of this slide (i.e. Randomization/Enrollment Date, Total Recruitment Goal, Rate of Enrollment per Month), we would have one program to create this exact same slide for any number of studies which in turn provides efficiency and standardization across all presentations made to DSMB members. The following code provides the macro code and an example of two different macro calls. Figure 7 shows the two different slides that are produced from two individual macro calls.

```sas
%MACRO ENROLLMENTFIGURE(SUBDS=, RAND=, RANDDT=, GOAL=, RATEPM=, FOOTNOTE=);
   /*Subject Level Dataset*/
   PROC SQL NOPRINT;
      SELECT MIN(&RANDDT) INTO: FIRSTRAND FROM &SUBDS;
   RUN;
```
Automated DSMB Presentation in SAS®: Yeah, I’d Submit That! How to auto populate PowerPoint presentations so you don’t have to, continued

DATA SUBS;
   SET &SUBDS;
   WHERE &RAND = 'Yes'; /*Randomization Indicator*/
   N = 1;
   MONTH = INTCK('MONTH', &FIRSTRAND, &RANDDT);
RUN;
PROC SORT DATA = SUBS;
   BY MONTH;
RUN;
PROC SQL NOPRINT;
   SELECT MAX(MONTH) INTO: LASTMONTH FROM SUBS;
RUN;

PROC FREQ DATA = SUBS NOPRINT;
   TABLES MONTH / OUT=MONTHS OUTCUM;
RUN;
DATA ACTUAL;
   SET MONTHS;
   GROUP = "Actual";
   RETAIN COUNT;
   SUBS=CUM_FREQ;
   KEEP MONTH SUBS GROUP;
RUN;
PROC SQL NOPRINT;
   SELECT PUT(MAX(SUBS), 3.) INTO: RANDN FROM ACTUAL;
   SELECT PUT(MAX(MONTH), 3.) INTO: LMON FROM ACTUAL;
RUN;
QUIT;
DATA PROJECT;
   GROUP = "Expected";
   PRO = 0;
   SUBSP=0;
   OUTPUT;
   PRO=(&GOAL./&RATEPM.);SUBSP=&GOAL.;OUTPUT;
RUN;
DATA PLOT1;
   SET ACTUAL PROJECT;
   LABEL GROUP = "Type: ";
RUN;
%MODSTYLE(NAME=STYLEONE, PARENT=JOURNAL, TYPE=CLM, COLORS=BLUE BLACK);
ODS POWERPOINT FILE = "&FILEPATH.\ENROLLMENTFIGURE.PPTX" NOGFOOTNOTE
STYLE=STYLEONE;
ODS GRAPHICS / WIDTH=8IN HEIGHT=8IN NOBORDER;
PROC SGPLOT DATA = PLOT1 NOAUTOLEGEND;
   SERIES X=PRO Y=SUBSP / LINEATTRS=(COLOR=BLACK) LEGENDLABEL="Expected"
   NAME="e";
   STEP X=MONTH Y=SUBSP / LINEATTRS=(COLOR=BLUE) CURVELABEL = "N= &RANDN. at Month &LMON." CURVELABELATTRS=(SIZE=9PT) LEGENDLABEL="Actual" NAME="a";
   XAXIS LABEL = "Study Month" OFFSETMAX=0.25;
   YAXIS LABEL = "Number Randomized" OFFSETMAX=.05;
   KEYLEGEND "e" "a" / NOORDER LOCATION=INSIDE POSITION=TOLEFT;
   TITLE "Enrollment By Study Month" BOLD H=35PT;
%IF &FOOTNOTE. ^= %THEN FOOTNOTE J=L "&FOOTNOTE";;
RUN;
ODS POWERPOINT CLOSE;
%MEND ENROLLMENTFIGURE;
Automated DSMB Presentation in SAS®: Yeah, I’d Submit That! How to auto populate PowerPoint presentations so you don’t have to, continued

```sas
%ENROLLMENTFIGURE (SUBDS=ADSL_100, RAND=RAND, RANDDT=RANDT, GOAL=100, RATEPM=10);

%ENROLLMENTFIGURE (SUBDS=ADSL_300, RAND=RAND, RANDDT=RANDT, GOAL=300, RATEPM=10);
```

**Enrollment By Study Month**

![Figure 7. Enrollment over Time Slides Produced from Two Separate Macro Calls](image)

**FIGURE SLIDE HELPFUL HINT**

If a figure that is created in a separate program is already sufficient for your presentation, or if a figure that will not be produced in SAS is needed in your presentation, you can still include the figure in your presentation. Some code to accomplish this can be found below:

```sas
/*Create a dummy dataset*/
DATA DUMMY;
  LABEL = "";
RUN;

/*Insert the figure using the PREIMAGE option in PROC REPORT*/
ODS LAYOUT GRIDDED COLUMNS=1;
ODS REGION WIDTH=7IN;
PROC REPORT DATA=DUMMY NOHEADER NOFS
  STYLE(REPORT)={PREIMAGE="<FIGURE PATH>"};
COLUMN LABEL;
  DEFINE LABEL / NOPRINT ;
RUN;
ODS LAYOUT END;
```

The figure must be in a non-PDF picture format (e.g. TIFF, PNG, JPG, etc.). The file may require resizing prior to including in the presentation to fit in a slide, which can be done after opening the file in Microsoft Paint.

**POPULATING A TEMPLATE**

In order to facilitate standardization across all studies reporting to the DSMB, we created a template PowerPoint presentation. To create our DSMB presentation template, we saved all macros that create all necessary slides in a single folder, then created a program (let’s call it DSMB_Slide_Macros.sas) that
used `%INCLUDE` statements to include each of the macros we needed. Here is a snippet of an example of the code to do this:

```sas
%INCLUDE "<Folder path>\Introduction_Slide_Macro.sas";
%INCLUDE "<Folder path>\Outline_Slide_Macro.sas";
%INCLUDE "<Folder path>\Study_Design_Slide_Macro.sas";
```

The program that holds all `%INCLUDE` statements was then included in each presentation program, followed by calls to each relevant macro bounded by an ODS POWERPOINT call. Using the above macros, some example code is shown here:

```sas
%INCLUDE "<Folder path>\DSMB_Slide_Macros.sas";
ODS POWERPOINT FILE="<Output Folder path>\DSMB_Presentation.sas";
  %Introduction_Slide(<Macro variable definitions>)
  %Outline_Slide(<Macro variable definitions>)
  %Study_Design_Slide(<Macro variable definitions>)
ODS POWERPOINT CLOSE;
```

Our template consists of 27 slides that we are using to generate DSMB presentations. Of those 27 slides, 19 are required while the other 8 are optional (as indicated in Table 1). Using the procedures that have been highlighted in the above examples, we created one program that creates all the slides needed for a DSMB presentation while facilitating certain levels of customization.

<table>
<thead>
<tr>
<th>Slide in DSMB Template Presentation</th>
<th>Study 1 1st DSMB 1 Site, Food Allergy Actively Enrolling</th>
<th>Study 2 2nd DSMB 3 Sites, Hay Fever Actively Enrolling</th>
<th>Study 3 4th DSMB 6 Sites, Atopic Dermatitis No Longer Enrolling</th>
</tr>
</thead>
<tbody>
<tr>
<td>Introduction</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Outline</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Study Design Text</td>
<td>X</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Study Design Figure</td>
<td>X</td>
<td></td>
<td></td>
</tr>
<tr>
<td>DSMB Recommendations</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Study Status</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Enrollment Table</td>
<td>X</td>
<td>X</td>
<td></td>
</tr>
<tr>
<td>Enrollment Table by Site - optional</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Enrollment Figure</td>
<td>X</td>
<td>X</td>
<td></td>
</tr>
<tr>
<td>Demographics Table</td>
<td>X</td>
<td>X</td>
<td></td>
</tr>
<tr>
<td>Subject Status Flow Chart</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Subject Status Table by Site - optional</td>
<td>X</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Subject Status Visits Bar Chart - optional</td>
<td>X</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Reasons for Early Termination Table</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Reasons for Early Termination Table by Site – optional</td>
<td>X</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Study Stopping Rules</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Study Stopping Rules Met</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Serious Adverse Events Summary Table</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>New Serious Adverse Events</td>
<td>X</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
Automated DSMB Presentation in SAS®: Yeah, I’d Submit That! How to auto populate PowerPoint presentations so you don’t have to, continued

<table>
<thead>
<tr>
<th>Slide in DSMB Template Presentation</th>
<th>Study 1 1st DSMB 1 Site, Food Allergy Actively Enrolling</th>
<th>Study 2 2nd DSMB 3 Sites, Hay Fever Actively Enrolling</th>
<th>Study 3 4th DSMB 6 Sites, Atopic Dermatitis No Longer Enrolling</th>
</tr>
</thead>
<tbody>
<tr>
<td>New Serious Adverse Events Narrative - optional</td>
<td>X</td>
<td></td>
<td>X</td>
</tr>
<tr>
<td>Non-Serious Adverse Event Summary Table</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Treatment and Procedure Related Adverse Events</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>WAO Grading of Treatment and Procedure Related Systemic Reactions - optional</td>
<td>X</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Epinephrine Use Table – optional</td>
<td></td>
<td></td>
<td>X</td>
</tr>
<tr>
<td>Protocol Deviation Table</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
<tr>
<td>Protocol Deviation Table by Site - optional</td>
<td>X</td>
<td></td>
<td>X</td>
</tr>
<tr>
<td>Appendices</td>
<td>X</td>
<td>X</td>
<td>X</td>
</tr>
</tbody>
</table>

Table 1. Example of slides required for various studies reporting to the same DSMB

The number of sites, therapeutic area, status of the study (i.e. enrolling, active with complete enrollment, closed to follow-up) and the number of times a study has been presented to the DSMB all play a role in determining the number of slides that need to be populated from the template.

Study 1 is a single site study that is presenting to the DSMB for the first time, is actively enrolling, and is in the Food Allergy therapeutic area. In contrast, Study 3 is a multi-site study that is presenting to the DSMB for the fourth time, is active but has completed enrollment, and is in the Atopic Dermatitis therapeutic area. These two studies are at very different points in the life cycle of a clinical trial and therefore provide the best opportunity to compare and contrast the macro calls that would be needed to populate our DSMB template. Table 2 shows 2 example programs containing macro calls that generate the set of slides needed for Study 1 and Study 3. For ease of comparison, we have removed the macro calls for the slides that both studies have in common.
Automated DSMB Presentation in SAS®: Yeah, I'd Submit That! How to auto populate PowerPoint presentations so you don't have to, continued

<table>
<thead>
<tr>
<th>Study 1 Program</th>
<th>Study 3 Program</th>
</tr>
</thead>
<tbody>
<tr>
<td>OPTIONS NODATE;</td>
<td>OPTIONS NODATE;</td>
</tr>
<tr>
<td>ODS POWERPOINT</td>
<td>ODS POWERPOINT</td>
</tr>
<tr>
<td>FILE=&quot;&amp;REPDIR.\Study_1_DSMB_Report.pptx&quot; NOGTITLE</td>
<td>FILE=&quot;&amp;REPDIR.\Study_3_DSMB_Report.pptx&quot; NOGTITLE</td>
</tr>
<tr>
<td>NOFOOTNOTE;</td>
<td>NOFOOTNOTE;</td>
</tr>
<tr>
<td>&lt;Introduction&gt;</td>
<td>&lt;Introduction&gt;</td>
</tr>
<tr>
<td>&lt;Outline&gt;</td>
<td>&lt;Outline&gt;</td>
</tr>
<tr>
<td>%Studydesign(DESIGN = A randomized trial in food allergy, ARM1 = Milk Consumption, ARM2 = Milk Avoidance, ENROLLGOAL = 100 randomized participants, PRIMARY = Proportion of subjects passing an oral food challenge at 52 weeks of age.)</td>
<td>%Studydesign(DESIGN = A randomized trial in atopic dermatitis, ARM1 = Active Therapy, ARM2 = Placebo, ENROLLGOAL = 300 randomized participants, PRIMARY = EASI Score at 4 weeks)</td>
</tr>
<tr>
<td>%EnrollmentTable(DSET = Study 1 Subject Level Dataset, LEVEL1LABEL = Screened, LEVEL1FL = SCREEN = 'Y', LEVEL2LABEL = Consented, LEVEL2FL = Consent = 'Y', LEVEL3LABEL = Randomized, LEVEL3FL = RAND = 'Y', GOALS = %QUOTE(N/A</td>
<td>200</td>
</tr>
<tr>
<td>%EnrollmentFigure(SUBDS = Study 1 Subject Level Dataset, RAND = Study 1 Randomization Indicator, RANDDT = Study 1 Randomization Date, GOAL = 100, RATEPM = 4)</td>
<td>&lt;Demographic Table&gt;</td>
</tr>
<tr>
<td>&lt;Demographic Table&gt;</td>
<td>&lt;Subject Status Flow Chart&gt;</td>
</tr>
<tr>
<td>...</td>
<td>...</td>
</tr>
<tr>
<td>&lt;Treatment and Procedure Related Adverse Events&gt;</td>
<td>&lt;Treatment and Procedure Related Adverse Events&gt;</td>
</tr>
<tr>
<td>%EpiUseSlide(EPIDS = Study 1 Epinephrine Use dataset, SUBDS = Study 1 Subject Level Dataset, IDVAR = Study 1 Subject Identifier, TRTAS = Study 1 Association with Treatment Flag, PROAS = Study 1 Association with Procedure Flag, AEAE = Study 1 Association with AE Flag, MOREONE = Study 1 More than One Epi Use Flag)</td>
<td>%NewSaeText(SAE1ID = &lt;ID of subject who had first SAE to display&gt;, SAE1EVENT = Headache, SAE1TEXT = 13 year old boy experienced headache at school 4 hours after treatment. The school nurse called 911. At the emergency department of the hospital, the physician did not see any signs of a headache. SAE resolved, and per protocol, the participant was discontinued from study treatment.)</td>
</tr>
<tr>
<td>%DevSlide(DEVDS = Study 1 Deviation Dataset, SUBDS = Study 1 Subject Level Dataset, IDVAR = Study 1 Subject Identifier, CAT = Study 1 Deviation category variable)</td>
<td>&lt;Non-Serious Adverse Event Summary Table&gt;</td>
</tr>
<tr>
<td>ODS POWERPOINT CLOSE;</td>
<td>ODS POWERPOINT CLOSE;</td>
</tr>
</tbody>
</table>

Table 2. Sample code for creating 2 DSMB presentations for differing studies
CONCLUSION

Using ODS POWERPOINT to create your DSMB presentations helps eliminate many opportunities for errors due to hand-entering values or copying and pasting tables and figures into your slides. Furthermore, by creating standardized, yet customizable, macros to accommodate many studies reporting to a DSMB, it is possible to create efficiencies that will save time and ensure DSMB presentations have the same formatting for every presentation. This paper explored several of the methods and tips that we used to create our DSMB presentation macros and will give you a start on creating your own.

ACKNOWLEDGMENTS

This work was performed as part of the Statistical and Clinical Coordinating Center for the National Institute of Allergy and Infectious Diseases (NIAID), Division of Allergy, Immunology, and Transplantation (DAIT) of the National Institutes of Health (NIH) grant UM2AI117870.

RECOMMENDED READING


CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Brett Jepson
Rho, Inc.
919-595-6239
Brett_Jepson@rhoworld.com

Kaitie Lawson
Rho, Inc.
919-595-6313
Kaitie_Lawson@rhoworld.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.